
Abstract

This package supplies convenient mechanisms for drawing flowcharts in
METAPOST. It includes commands for drawing line shapes (e.g., rectangles,
ovals, drums, etc.), text labels, fill patterns, and arrow connectors.

1 Introduction

METAPOST is a superior means of drawing scientific diagrams for LATEX docu-
ments for the following reasons:

• Precision: METAPOST graphics have a very clean, precise look because
designers mathematically specify the correct placement of all figure elements
rather than estimating their locations by point-and-click.

• Scalability: METAPOST outputs pure vector graphics that exhibit no quality
degredation with scaling, making them ideal for professional publishing.

• Output compatability: Using graphicx, LATEX can import METAPOST

graphics directly into DVI, PostScript, and PDF documents without con-
verting them to a lower quality graphic format.

• Small size: Embedded METAPOST graphics are typically much smaller than
alternative formats, making the resulting documents more convenient to
serve over the web.

The METAf low package creates convenient METAPOST macros for drawing
flowcharts and other line-art pictures. Unlike other similar METAPOST packages,
METAf low infers all shape positions and sizes from linear constraints rather than
requiring the user to specify them as explicit macro parameters. This leverages the
considerable power of METAPOST’s constraint-solver to position and size shapes
in natural ways, such as by auto-sizing them to their labels or auto-positioning
them relative to other shapes.

1 input metaflow

2 prologues := 2;

3 filenametemplate "%j-%c.mps";

4 beginfig(1)

5 z1c = (0,0);

6 draw rect1(btex experiment etex);

7 putitem2 20right of 1;

8 draw oval2(btex results etex);

9 drawarrow connector(1,2,right,right);

10 putitems(2,3) like (1,2);

11 z3s = (55,30);

12 draw diamond3(btex evaluate etex);

13 drawarrow connector(2,3,right,right);

14 drawarrow connector1(3,1,down,up);

15 z4um = point 1.5 of cp1;

16 drawopen rect4(btex revise etex);

17 putitems(3,5) like (1,2);

18 drawopen rect5(btex success etex);

19 drawarrow connector(3,5,right,right);

20 endfig;

21 end

(a) Listing of chart.mp

1 \documentclass{article}

2 \usepackage{graphicx}

3 \begin{document}

4 Here is a nice diagram:

5 \begin{center}

6 \includegraphics

7 [width=.7\hsize]

8 {chart-1.mps}

9 \end{center}

10 \end{document}

(b) Listing of sample.tex

experiment results evaluate

revise

success

(c) The image that results from the two listings above

Figure 1: A simple METAPOST program and the image it draws

2 Basic Usage

The METAPOST and LATEX source files in Figs. 1(a) and 1(b) produce the
flowchart in Fig. 1(c) using METAf low . To compile the sample, perform the fol-
lowing steps:

1. Create a new text file named chart.mp with the content of Fig. 1(a).

2. Copy the metaflow.mp and mftext.tex files into the same directory.

3. Run METAPOST: mpost -tex=latex chart.mp

4. Create a new text file named sample.tex with the content of Fig. 1(b).

5. Run LATEX: pdflatex sample.tex

Initialization. Line 1 of Fig. 1(a) loads the metaflow package; it requires the
package file metaflow.mp to be in the same directory as your chart.mp file. Line 2
asks METAPOST to add font metric information to the output graphics, which is
necessary for compatibility with many DVI and PostScript viewers.

Line 3 specifies that each output file should be named 〈jobname〉-〈number〉.mps.
An mp file may contain many figures, each of which begins with beginfig (Line 4)
and ends with endfig (Line 20). The number in the beginfig line determines
the 〈number〉 part—in this case chart-1.mps.

Anchor points. Line 5 defines a point named z1c located at the origin. Point
names in METAPOST start with prefix z, followed by a numerical index (e.g.,
1), and concluding with an alphabetic suffix (e.g., c). The METAf low package
reserves certain suffixes for anchor points of shapes, as illustrated in Fig. 2. Suffix
c is for the center point of the shape, so Line 5 states that the center of shape 1 is
at the origin. To refer to the x- or y-value of a point, just use prefix x or y in place
of z. For example, we could have instead written x1c=0 and y1c=0 separately.

Defining the position of any anchor point defines the position of the whole
shape. Positions can also be expressed relative to other points. For example,

z2ml = z1mr + (20,0); (1)

says that the middle-left (ml) point of shape 2 is 20 points to the right and 0
points above the middle-right (mr) point of shape 1.

The putitem macro makes such constraints easier to type. To place the edgeputitem

of shape 〈i〉 a distance 〈n〉 in the 〈dir〉 direction from shape 〈j 〉, write

putitem〈i〉 〈n〉〈dir〉 of 〈j 〉

where 〈dir〉 is one of up, down, left, right, upright, downright, upleft, or
downleft. For example, Line 7 of Fig. 1(a) is equivalent to statement (1) above.

To “copy” the relative positioning of a pair of items, use the putitems macro:putitems

putitems(i1,j1) like (i2,j2);

applies the same putitem command to items i1 and j1 as was applied to position
items i2 and j2. For example, Line 10 says that items 2 and 3 should be relatively
positioned like items 1 and 2 (as specified in Line 7). This is better than retyping
the “20right” in Line 7 because it allows you to later fine-tune the placement of
all the figure items by changing only Line 7 instead of all its copies.

Sizes. Suffix s is reserved for the size of shapes. For example, Line 11 says that
shape 3 is 55 points wide and 33 points high.

Whenever a shape has a label, METAf low assigns a default size to suffix ds.
You can specify the shape’s size relative to this default by writing constraints like

z〈i〉s = z〈i〉ds + (5,2);

This makes shape 〈i〉 5 points wider and 2 points higher than its default size. If
you do not specify a shape’s size and the shape has a label, METAf low uses the
default size. If it does not have a label, you must specify a size.

Some combinations of constraints make it unnecessary to explicitly specify a
size; for example, if you specify the positions of the lower-left and upper-right
corners, METAf low infers the resulting size automatically.

Shapes. The draw command draws a shape at its prespecified position. The
various shapes, their names, and their anchor points are shown in Fig. 2. Labels,
if provided, are centered within the shape. To leave the shape unlabeled, you can
omit the label (leaving an empty pair of parentheses). Be sure to specify a size in
this case (see above).

Using drawopen instead of draw draws a shape’s label (and any fills) withoutdrawopen

drawing its border. This is convenient for drawing text boxes, as demonstrated
by Lines 16 and 18 of Fig. 1(a).

Connectors. The expression connector(i1,i2,dir1,dir2) returns a connectorconnector

path from shape i1 to shape i2. The path leaves shape i1 in direction dir1 and
enters shape i2 in direction dir2. Indexes i1 and i2 are numbers, and directions
dir1 and dir2 are each one of up, down, left, or right. The path avoids passing
through shapes i1 and i2, but does not attempt to avoid any other shapes.

To draw a connector path with an arrow at the end, use the drawarrow com-
mand, as in Lines 9, 13, and 19. To draw the path without an arrowhead, just use
draw. To draw arrowheads at both ends, use drawdblarrow.

Line 14 assigns the name “1” to the connector it draws. This allows Line 15 to
use the expression “point n of cp1” to refer to the nth point along connector
path 1. In general, the 0th point is the start point, the nth point is the nth bend
in the path, and the last point is the end point. Fractional points are interpolated,
so the 1.5th point is halfway between the 1st and 2nd points.

Line styles and colors. Any METAPOST drawing options can be used at the
end of any kind of draw command to specify the line style and color. Figure 3
shows some examples. These can come at the end of any draw command, including
drawarrow, and can be combined when relevant. For example,

draw rect1() withpen (pencircle scaled 4)
withcolor (red)
dashed (evenly scaled 4);

draws a rectangle whose border is a thick, red, long-dashed line.

Filled shapes. Shapes can be filled with solid colors, stripes, or tesselated pat-
terns by using the filledwith, stripedwith, and tesselatedwith operators,
respectively. Figure 4 illustrates each. When line styles and fill styles are com-
bined, as in Fig. 4(d), all line styles must come after all fill operators.

4



urumul

ml

ll lm lr

mrc

(a) rect

urumul

ml

ll lm lr

mrc

(b) rrect

ur
um

ul

ml

ll
lm

lr

mrc

(c) oval

urumul

ml

ll lm lr

mrc

(d) rhomb

urumul

ml

ll lm lr

mrc

(e) trap

ur

um

ul
ml

ll

lm

lr

mrc

(f) diamond

ur
um

ul

ml

ll
lm

lr

mrc

(g) circ

ur
um

ul

ml

ll lm lr

mrc

(h) drum

urumul

ml

ll lm lr

mrc

(i) tornbox

Figure 2: Shapes and their anchor points

draw. . . dashed evenly;

draw. . . dashed evenly scaled 4;

draw. . . dashed withdots;

draw. . . withpen pencircle scaled 4;

draw. . . withpen pencircle scaled 4 withcolor .5white;

draw. . . withpen pencircle scaled 4 withcolor red;

Figure 3: Line style examples

(a) 1 draw rect1() filledwith .5white;

(b)
2 draw oval2()

3 stripedwith evenstripes scaled 2 rotated 30 colored .7white;

(c)

4 picture polkadots;

5 polkadots = image(fill fullcircle scaled 6 withcolor .5white;

6 fill fullcircle scaled 6 shifted (6,6)

7 withcolor .5white;);

8 draw trap3() tesselatedwith polkadots;

(d)

9 draw rhomb4()

10 stripedwith evenstripes colored .7white

11 stripedwith pinstripes rotated 90 colored .7white

12 withpen pencircle scaled 2 dashed evenly scaled 2;

Figure 4: Filled shapes

The filledwith operator fills shapes with solid colors, as seen in Fig. 4(a).filledwith

The stripedwith operator fills shapes with stripe patterns, as demonstratedstripedwith

evenstripes

pinstripes

by Fig. 4(b). A stripe pattern is usually the predefined picture evenstripes

optionally scaled to change stripe thickness, optionally rotated to change ori-
entation, and optionally shifted to adjust position. The alternative pinstripes

picture creates stripes of zero thickness (i.e., lines).
Any picture’s color can be adjusted with the colored operator. This is usefulcolored

for changing stripe colors, as shown in Figs. 4(b) and 4(d). Note that the colored

operator is not the same as the withcolor modifier; the former changes the color
of a picture (e.g., a fill) whereas the latter specifies a line color for a draw command.

The tesselatedwith operator fills shapes with a tesselated, rectangular pic-tesselatedwith

ture. Figure 4(c) constructs such a picture using METAPOST’s image macro.
Multiple fill operators can be layered, as in Fig. 4(d), to produce cross-hatching

or other effects. They are applied from first to last, with opaque parts of later fills
occluding those that came before. As noted earlier, any line style options must
come last, after all fill operators, as Line 12 demonstrates.

Layering. By default, each drawing command contributes new material on top,inback

occluding anything that may have been drawn previously (except that item labels
are always drawn atop everything else). To instead draw something underneath
what has been drawn before, prefix it with the word inback:

inback draw rect3() filledwith .8white;

More complex layering can be achieved with the stand-alone commandturntolayer

turntolayer(〈n〉), which causes all subsequent drawing commands until the next
turntolayer command to draw onto layer number 〈n〉. The starting layer is num-
ber 0, with lower-numbered layers drawn behind higher-numbered ones. Negative-
numbered layers are permitted.

3 Text

METAPOST documents typeset textual content in LATEX using btex 〈text〉 etex:

draw rect1(btex Approximate $f^2(x)$ etex)

To use LATEX (rather than plain TEX) to typeset such text, perform 3 steps:

1. Copy the included mftext.tex helper library to your working directory.

2. Add the following code to your mp file somewhere before your first figure:

verbatimtex

\documentclass[10pt]{article}

\renewcommand\familydefault{\sfdefault}

\input mftext

\begin{document}

etex

6



This material, if used, must be placed directly in the top-level mp file (not in
an auxilliary file included via input), since that is the only place METAPOST
looks for it.

3. Execute METAPOST with: mpost -tex=latex 〈mpfile〉

The LATEX code in step 2 accomplishes three things:

• It sets up a LATEX environment rather than one limited to plain TEX.

• It makes 10-point Sans Serif the default font.

• It introduces macros \textc, \textl, and \textr, which center-, left-, and
right-align (respectively) multiline text separated by \\ (see Fig. 5).

this is
left-aligned
text

draw rect1(btex \textl{this is \\ left-aligned \\ text} etex);

this is
center-aligned

text
draw rect2(btex \textc{this is \\ center-aligned \\ text} etex);

this is
right-aligned

text
draw rect3(btex \textr{this is \\ right-aligned \\ text} etex);

Figure 5: LATEX labels with textual alignment

4 Advanced Features

4.1 Rotated Shapes

Shapes can be rotated using the rshape operator:

rshape〈i〉(〈shape〉,〈dir〉)(〈label〉)

Figure 6 illustrates by rotating a trapezoid. Direction 〈dir〉 is one of up, down,
left, or right, where the up direction leaves the shape upright (i.e., unrotated).

ur
um

ul

ml

ll
lm

lr
mrc draw rshape1(trap,right)();

Figure 6: A rotated trapezoid

a b

c

d

1 z1a = (0,0);

2 z1b = (70,0);

3 filldraw brace1(up);

Figure 7: Braces

4.2 Braces

Curly braces are created as in Fig. 7. Define at least the a and b anchor points
(which are interchangeable), and indicate the direction of the brace in the macro’s
sole argument. The height and position of the cusp can be adjusted by additionally
defining the c anchor point. The position but not the height can be adjusted by
defining anchor point d instead, which must lie on the line segment from a to b.

The brace macro returns an outline path for the brace, so typically one should
use filldraw to draw the outline and fill it. Unlike other shapes, braces do not
have a size (s), default size (ds), or built-in label; and they rotate to match the
slope of the line a–b.

4.3 Shape Adjustments

Most variables in METAPOST are immutable—their values never change. For
example, the constraint “z1c=(0,0)” says that the center of shape 1 is the origin
forever. In contrast, the variables described in this section are mutable—their
values may change, andMETAf low uses the current value when defining shapes. To
change the value of a mutable parameter, use the special assignment operator “:=”.
For example, command “rradius:=10” changes the radii of of future rounded
rectangle corners to 10 (see below).

The radii of the rounded corners of rrect shapes can be changed by reassigningrradius

rradius.
The angle of the bottom-left corner of a rhomb or trap shape is given byrhombangle

rhombangle, which must be a number between 0 and 180.
The ratio of the height to width of a drum’s lid is given by drumlidratio. Itsdrumlidratio

default value is 0.2.
When sizing shapes to fit their labels, the minimum distance permitted betweenilmargin

the item label and its border is given by the number ilmargin.
Figure 8 illustrates the six parameters that control the appearance of curlybraceheight

clawwidth

beekheight

beekangle

bracethick

braceret

braces yielded by brace, along with their default values. The braceheight pa-
rameter is merely a default that is only used when the cusp anchor point (c) for
the brace is not pre-specified.

4.4 Custom Connectors

Declaring a named connector via connector〈i〉( . . . ) introduces an array of points
named z〈j 〉cp〈i〉 where 〈i〉 identifies the connector and 〈j 〉 identifies a bend or
endpoint of the connector. For example, z0cp5 is the start point of connector 5,
z1cp5 is its 1st bend (or endpoint if it has no bends), etc. Pre-specifying values for

beekangle (10◦)

bracethick (1.5)

braceret (1)
clawwidth (12)

beekheight (4.5)

b
r
a
c
e
h
e
i
g
h
t
(1
0)

Figure 8: Brace parameters

1 2

1 drawarrow connector(1,2,right,left);

2 x1cp8 = .75[x1mr,x2ml];

3 drawarrow connector8(1,2,right,left) dashed evenly;

4 y2cp9 = y2um + 6;

5 drawarrow connector9(1,2,right,left) dashed evenly scaled 2;

Figure 9: Custom connectors

these points before the connector operator is used has the effect of customizing
the connector path.

Figure 9 demonstrates. Line 1 draws the default connector path (the solid line)
departing rightward from box 1 and entering leftward into box 2. Line 2 asserts
that the x-ordinate of bend 1 of connector path 8 (x1cp8) should be 75% of the
way from the middle-right of box 1 (x1mr) to the middle-left of box 2 (x2ml). This
results in the short-dashed connector path in the figure. (Note that most of the
path overlaps the solid-line default path and cannot be seen.) Line 4 asserts that
the y-ordinate of bend 2 of connector path 9 (y2cp9) should be 6 points above
the y-ordinate of the upper-middle point of box 2 (y2um). This causes the path to
loop overtop box 2 instead of underneath, resulting in the long-dashed connector
path in the figure.

METAf low will never change the number of bends in a path in response to
connector customizations. If you want to radically change the path strategy, you
should draw your own path from scratch using METAPOST commands instead of
using the connector operator.

Default connector paths avoid passing within cmargin points of the boundingcmargin

boxes of the source and destination items. You can adjust this margin by changing
the value of cmargin:

cmargin := 10;

The popover macro can be used to “pop” one connector path over its inter-popover

pradius sections with a list of other paths, as demonstrated by Fig. 10. The syntax

〈path〉 popover(〈path list〉)

1

23

4

5

6

1 path p;

2 p = z1lr--z2ul;

3 drawarrow p;

4 drawarrow connector1(3,4,up,right);

5 drawarrow connector(5,6,right,up) popover(p,cp1);

Figure 10: Popovers

1 picture mystripes;

2 mystripes = dashstripes dashpattern(on 5 off 3 on 1 off 3);

3 draw oval1() stripedwith mystripes;

4 picture s;

5 s = dashstripes

6 dashpattern(onhue(red) 3 onhue(green) 3 onhue(blue) 3);

7 draw oval2() stripedwith s rotated -45;

Figure 11: Custom stripe patterns

returns a path in which semi-circular arcs have been spliced into 〈path〉 wherever
it intersects any of the paths in 〈path list〉 (a comma-separated list of paths). To
change the radii of the arcs, modify pradius (e.g., “pradius:=5”). Any intersec-
tions closer than pradius to the ends of the 〈path〉 or from any other intersections
are ignored by popover.

4.5 Custom Stripe Patterns

In addition to the predefined evenstripes and pinstripes stripe patterns,dashstripes

authors may define their own stripe patterns by first defining a METAPOST

dashpattern and then converting it to a stripe pattern with the dashstripes

operator. For example, Line 2 of Fig. 11 creates a stripe pattern consisting of a
long (5-point) dash, a 3-point gap, then a short (1-point) dash, another 3-point
gap, repeating.

The dashstripes operator projects each dash orthogonally to form a stripe.
Since METAPOST dashpatterns are horizontal, this means that custom stripes
start out vertical. To rotate them, apply the rotated operator to the result of
the dashstripes operation.

A dash that has zero width (created via “on 0” in the dashpattern argument)
becomes a line when striped, like the stripes in the pinstripes pattern. A pat-
tern consisting solely of pinstripes must have more than one in the dashpattern

operand. For example, instead of writing dashpattern(on 0 off 3), which con-
sists of exactly one pinstripe and is therefore illegal, write dashpattern(on 0 off

3 on 0 off 3), which is equivalent but has two pinstripes, satisfying the require-
ment. (A pattern consisting of exactly one pinstripe is not permitted because a
zero-width dash is a directionless point, preventing dashstripes from identifying

(a) ahinset := 0 (b) ahinset := .4 (c) ahinset := 1

Figure 12: Arrowhead variants

the direction orthogonal to the pattern.)
Custom stripe patterns can be recolored using colored in the typical wayonhue

(see §2), but multicolored stripe patterns can be created directly using the onhue

dash pattern operator. Line 6 demonstrates. Operation onhue(〈color〉) is like on

except that it additionally specifies the 〈color〉 of the dash.

4.6 Helper Variables and Macros

Defining a shape item introduces a new path variable ip〈n〉 (where 〈n〉 is theip

item’s name) that holds the border of the shape. This is useful if you want to use
METAPOST commands and operations to find (non-anchor) points on the item’s
border. The borders of all shapes other than drums are cycles.

The picture variable il〈n〉 holds the label (if any) of item 〈n〉, and variableil

ls z〈n〉ls holds its size.
Each named custom connector declared via connector〈i〉(. . .) (see §4.4) intro-cp

duces a variable named cp〈i〉 that stores the connector path, and variables named
z〈j 〉cp〈i〉 for the jth endpoint or bend in the path.

Plain METAPOST provides llcorner, urcorner, lrcorner, and ulcornerllcorners

urcorners

lrcorners

ulcorners

primitives that return the lower-left, upper-right, lower-right, and upper-left cor-
ner of a path or picture. METAf low extends these with plural forms that return
the corners of a list of paths, pictures, and/or points. For example, the following
creates a rectangle that circumscribes items 1, 2, and 3 (definitions are not shown).

z4ll = llcorners(ip1,ip2,ip3) - (10,10);

z4ur = urcorners(ip1,ip2,ip3) + (10,10);

draw rect4();

4.7 Arrowheads

The back edge of arrowheads can be customized by adjusting the value of ahinset
to a value between 0 and 1. The possibilities are illustrated in Fig. 12. The
default value of 0 draws arrowheads with straight back edges, value 1 draws open
arrowheads with no back edge, and values between 0 and 1 yield V-shaped back
edges.

Rectangle 1

Rect 2

1 z1c = origin;

2 z1lm = z2um;

3 rect1(btex Rectangle 1 etex,false);

4 rect2(btex Rect 2 etex,false);

5 z1s = z2s = (max(x1ds,x2ds), max(y1ds,y2ds));

6 draw rect1();

7 draw rect2();

Figure 13: Rectangles with interdependent parameters

4.8 Interdependent Shapes

Sometimes the parameters of two or more shapes interrelate in such a way that
none can be finalized and drawn until the others are declared. For example,
suppose rectangles 1 and 2 should have identical sizes that are the maximum of
their respective default sizes (as determined by their labels). Maximization is a
non-linear function, so it cannot be specified as a linear constraint in METAPOST.
Both default sizes must therefore be known before the size of either shape can be
computed.

Figure 13 illustrates how this can be accomplished in METAf low with three
steps. First, declare each shape without finalizing or drawing it by supplying the
optional boolean argument false to the shape’s constructor (Lines 3–4). Second,
supply any interdependent or non-linear constraints necessary to resolve all un-
knowns for the shapes (Line 5). Third, finalize and draw the shapes by applying
the shape constructors again with no arguments (Lines 6–7).

4.9 Scripting

When drawing, it is convenient to have a means of quickly inspecting the results
of edits. On Unix I recommend creating a Makefile with the following content:

texfiles = mydocument.tex

mpfiles = myfigures.mp

all: $(texfiles:%.tex=%.pdf) $(mpfiles:%.mp=%-1.mps)

%.pdf: %.tex $(mpfiles:%.mp=%-1.mps)
-|pdflatex $<
-|if grep "may have changed" $*.log; then pdflatex $<; fi

%-1.mps: %.mp
-|mpost -tex=latex $<
-|touch $@

where mydocument.tex and myfigures.mp are the names of your main .tex and
.mp files, respectively. Then execute gmake to compile all documents and figures.

On Windows I recommend creating a plain text file (e.g., with Notepad) named
makefigs.bat in the same directory as your tex and mp files with the following
content:

mpost -tex=latex myfigures.mp

@IF ERRORLEVEL 1 PAUSE & EXIT

pdflatex mydocument.tex

@PAUSE & EXIT

where mydocument.mp and myfigures.tex in the first and third lines should be
replaced with the filenames of your mp and tex file, respectively. Double-click
on your makefigs.bat file to recompile your document, including recompiling all
figures.

5 Implementation

This package requires at least version 1.004 of METAPOST, since that version
introduced the colorpart macro. (Earlier versions had broken or non-existent
color primitives.)

1 if unknown mpversion: errmessage

2 "MetaPost v1.004 or later required (found one older than v0.900)";

3 elseif scantokens(mpversion) < 1.004: errmessage

4 "MetaPost v1.004 or later required (found v" & mpversion & ")";

5 fi

init_metaflow Each new figure is initialized by declaring ip, il, and cp as variable classes for
item paths, item labels, and connector paths, respectively. The layer array is
also initialized.

6 def init_metaflow =

7 save ip, il, cp, layer;

8 path ip[], cp[];

9 picture il[], layer[];

10 thislayer := 0;

11 layerlist := origin;

12 itemlabels := nullpicture;

13 enddef;

14 extra_beginfig := extra_beginfig & "init_metaflow;";

At the end of each figure, draw all the item labels and layers. Drawing labels
at the end prevents them from being covered by fills.

15 extra_endfig := extra_endfig & "flatten;";

rradius The rounded corners of an rrect shape have radii rradius.

16 newinternal rradius;

17 rradius := 5;

pradius The popover macro introduces “pops” of radius pradius.

18 newinternal pradius;

19 pradius := 3;

cmargin Connector paths avoid passing within distance cmargin of the bounding box of
the source or destination shapes. By default we set this to 1.5 times the length of
an arrowhead. This prevents bends within connector arrowheads.

20 newinternal cmargin;

21 cmargin := 1.5ahlength;

ilmargin When sizing shapes based on textual labels, the minimum distance from the label
to the shape edge in the vertical and horizontal directions is dictated by ilmargin.

22 newinternal ilmargin;

23 ilmargin := 3;

drumlidratio Set the default ratio of drum lid height to width.

24 newinternal drumlidratio;

25 drumlidratio := .2;

rhombangle Set the bottom-left angle of rhomboid shapes in degrees. This parameter must be
a value between 0 and 180.

26 newinternal rhombangle;

27 rhombangle := 80;

braceheight

clawwidth

beekheight

braceret

beekangle

bracethick

The following parameters control the appearance of curly braces drawn with the
brace macro as illustrated in Fig. 8. Specifically,

• braceheight is the default height of the c anchor point above the a–b line,

• clawwidth is the width of the curved end pieces (and also the symmetric
curves at the cusp),

• beekheight is the distance from the cusp to an imaginary line that passes
through most of the brace,

• braceret is the vertical distance between the hill and valley of the curve,

• beekangle controls the pointiness of the cusp, and

• bracethick is the thickness of the curve at its thickest points.

28 newinternal braceheight, clawwidth, beekheight;

29 newinternal braceret, beekangle, bracethick;

30 braceheight := 10;

31 clawwidth := 12;

32 beekheight := 4.5;

33 braceret := 1;

34 beekangle := 10;

35 bracethick := 1.5;

layerlist

thislayer

The list of layers is stored as a sorted path of integer points along the major
diagonal. Its value is reinitialized at the start of each figure, and updated with
each layer-change.

36 path layerlist;

37 newinternal thislayer;

itemlabels A picture consisting of all item labels is accumulated separately from the accumu-
lated picture so that all labels can all be drawn together at the end of the figure.
This prevents fills from covering labels.

38 picture itemlabels;

itemfinal Users may optionally suppress the final portion of each shape macro, allowing
constraints to remain unresolved and delaying the construction of the shape path
and the drawing of the label. The following boolean remembers whether we’re
finalizing the current item now.

39 boolean itemfinal;

gensuf The following macro was adapted from the generisize macro in boxes.mp. It
takes a string version of a suffix (as returned by str) and returns a new string in
which all explicit numeric subscripts have been replaced by generic brackets ([]).
Shapes that are presented with non-standard names use this to declare the types
of new variables that have the non-standard name as a suffix.

40 vardef gensuf(expr s) =

41 save n,r,c; string r,c;

42 n := 0; r := "";

43 forever: exitunless n < length s;

44 c := substring(n,n+1) of s;

45 if (c>="0") and (c<="9"):

46 r := r & "[]";

47 forever: n := n + 1;

48 c := substring(n,n+1) of s;

49 exitunless (c=".") or ((c>="0") and (c<="9"));

50 endfor

51 elseif c="[":

52 if (substring(n+1,n+2) of s)="[":

53 r := r & "[["; n := n + 2;

54 else:

55 r := r & "[]"; n := n + 1;

56 forever: exitunless n < length s;

57 n := n + 1;

58 exitif (substring(n-1,n) of s)="]";

59 endfor

60 fi

61 else:

62 r := r & c; n := n + 1;

63 fi

64 endfor

65 r

66 enddef;

inititem Initialize a new shape item. All shapes must be named, since at least their po-
sitions must be pre-specified, and there is no way to do that without a name.
Initialization involves defining the bounding box constraints that are common to
all shapes, and parsing any arguments (e.g., the optional label).

67 vardef inititem@#(text _t) =

68 if (str @#)="":

69 errmessage("unnamed shape");

70 fi;

71 if known ip@#:

72 errmessage("redundant shape name: " & (str @#));

73 fi;

74 z@#bb.c = z@#c;

75 z@#bb.ur = z@#c + .5z@#s;

76 z@#bb.ll = z@#c - .5z@#s;

77 z@#bb.ul = (x@#bb.ll, y@#bb.ur);

78 z@#bb.lr = (x@#bb.ur, y@#bb.ll);

79 z@#bb.um = .5[z@#bb.ul,z@#bb.ur];

80 z@#bb.lm = .5[z@#bb.ll,z@#bb.lr];

81 z@#bb.ml = .5[z@#bb.ll,z@#bb.ul];

82 z@#bb.mr = .5[z@#bb.lr,z@#bb.ur];

83 save _pic, _fin; picture _pic; boolean _fin;

84 _fin := true;

85 for __t=_t:

86 if picture __t:

87 _pic = __t;

88 elseif string __t:

89 _pic = __t infont defaultfont scaled defaultscale;

90 elseif boolean __t:

91 _fin := __t;

92 else:

93 errmessage("illegal shape argument type");

94 fi

95 endfor;

96 itemfinal := _fin;

97 if known _pic:

98 z@#ls = urcorner _pic - llcorner _pic;

99 if not picture il@#:

100 scantokens ("picture il." & gensuf(str @#));

101 fi

102 il@# = _pic;

103 fi

104 enddef;

finitem Finish an item by drawing its optional label, defining its frame path (if there is
a properly typed variable to receive it), and returning the frame path so that a
drawing command can draw it. Rather than contributing the label directly to
the current picture, it is drawn into a separate labelitems picture that will be
added to the overall picture at the end. This prevents labels from being covered
by fills. The path is only stored in a variable if doing so would not cause an error.
This allows users to define items with non-standard names without first declaring
a path variable when the path variable is never used.

105 vardef finitem@#(text p) =

106 if itemfinal:

107 if unknown x@#s: (x@#s,0) = (x@#ds,0) fi;

108 if unknown y@#s: (0,y@#s) = (0,y@#ds) fi;

109 if known il@#:

110 addto itemlabels also

111 (il@# shifted (z@#lc-.5[llcorner il@#,urcorner il@#]));

112 fi

113 if (path ip@#) and (unknown ip@#): ip@#=p; ip@# else: p fi

114 fi

115 enddef;

The following macros define shapes. Each shape definition begins with a call
to inititem, then introduces constraints that tie all anchor points to the bound-

ing box points (z@#bb. . .), then finishes the shape with a call to finitem. This
ordering is important because it maximizes the chances that constraints can be
resolved prior to reaching operations that fail for unresolved constraints.

Whenever an item label is given, each shape defines a default size z@#ds based
entirely on the label size z@#ls. Some shapes require this relationship to be non-
linear; in that case default size constraints are only computed when the label size
is fully known.

rect Define a rectangular item.

116 vardef rect@#(text cap) =

117 inititem@#(cap);

118 z@#lr = z@#bb.lr;

119 z@#ur = z@#bb.ur;

120 z@#ul = z@#bb.ul;

121 z@#ll = z@#bb.ll;

122 z@#lm = z@#bb.lm;

123 z@#mr = z@#bb.mr;

124 z@#um = z@#bb.um;

125 z@#ml = z@#bb.ml;

126 z@#lc = z@#c;

127 z@#ds = z@#ls + (2ilmargin,2ilmargin);

128 finitem@#(z@#ll--z@#lr--z@#ur--z@#ul--cycle)

129 enddef;

rrect Define a rounded rectangular item.

130 vardef rrect@#(text cap) =

131 inititem@#(cap);

132 z@#lm = z@#bb.lm;

133 z@#mr = z@#bb.mr;

134 z@#um = z@#bb.um;

135 z@#ml = z@#bb.ml;

136 z@#ll-z@#bb.ll = z@#bb.ur-z@#ur = rradius*(1-sqrt(.5))*(1,1);

137 z@#lr-z@#bb.lr = z@#bb.ul-z@#ul = rradius*(1-sqrt(.5))*(-1,1);

138 z@#lc = z@#c;

139 z@#ds = z@#ls + 2*(if (rradius-ilmargin)*sqrt(2) > rradius-1:

140 (rradius-(rradius+1)/sqrt(2))*(1,1)

141 else: (ilmargin,ilmargin) fi);

142 finitem@#(

143 (subpath (0,2) of fullcircle scaled 2rradius

144 shifted (z@#bb.ur-(rradius,rradius)))--

145 (subpath (2,4) of fullcircle scaled 2rradius

146 shifted (z@#bb.ul+(rradius,-rradius)))--

147 (subpath (4,6) of fullcircle scaled 2rradius

148 shifted (z@#bb.ll+(rradius,rradius)))--

149 (subpath (6,8) of fullcircle scaled 2rradius

150 shifted (z@#bb.lr-(rradius,-rradius)))--

151 cycle

152 )

153 enddef;

_rax This helper macro safely computes the x that satisfies x/y = tan θ where y is given
and θ is rhombangle.

154 vardef _rax(expr y) =

155 save ?; numeric ?;

156 (?,y) = whatever * dir rhombangle;

157 ?

158 enddef;

rhomb Define a rhomboid item.

159 vardef rhomb@#(text cap) =

160 inititem@#(cap);

161 z@#lm = z@#bb.lm;

162 z@#mr = .5[z@#lr,z@#ur];

163 z@#um = z@#bb.um;

164 z@#ml = .5[z@#ll,z@#ul];

165 z@#bb.ur-z@#ur = z@#ll-z@#bb.ll = (whatever,0);

166 z@#bb.ul-z@#ul = z@#lr-z@#bb.lr = (whatever,0);

167 z@#lc = z@#c;

168 z@#ul-z@#ll = whatever * dir rhombangle;

169 if rhombangle<90: z@#ll = z@#bb.ll

170 else: z@#ul = z@#bb.ul fi;

171 if known y@#ls:

172 z@#ds = z@#ls + 2*(abs(_rax(y@#ls+2ilmargin)) +

173 max(ilmargin-abs(_rax(ilmargin)),0),

174 ilmargin);

175 fi

176 finitem@#(z@#ll--z@#lr--z@#ur--z@#ul--cycle)

177 enddef;

trap Define a trapezoid item.

178 vardef trap@#(text cap) =

179 inititem@#(cap);

180 z@#lm = z@#bb.lm;

181 z@#mr = .5[z@#lr,z@#ur];

182 z@#um = z@#bb.um;

183 z@#ml = .5[z@#ll,z@#ul];

184 z@#ul-z@#bb.ul = z@#bb.ur-z@#ur = (whatever,0);

185 z@#ll-z@#bb.ll = z@#bb.lr-z@#lr = (whatever,0);

186 z@#lc = z@#c;

187 z@#ul-z@#ll = whatever * dir rhombangle;

188 if rhombangle<90: z@#ll = z@#bb.ll

189 else: z@#ul = z@#bb.ul fi;

190 if known y@#ls:

191 z@#ds = z@#ls + 2*(abs(_rax(y@#ls+2ilmargin)) +

192 max(ilmargin-abs(_rax(ilmargin)),0),

193 ilmargin);

194 fi

195 finitem@#(z@#ll--z@#lr--z@#ur--z@#ul--cycle)

196 enddef;

diamond Define a diamond item. The default diamond size is chosen to be the one that
minimizes the sum a+b while still circumscribing the label, where a and b are half
the width and height of the diamond, respectively. If a and b are both free, then
it turns out that the optimal diamond satisfies a = x +

√
xy and b = y +

√
xy,

where (x, y) is the upper right corner of the label when it is centered at the origin.

197 vardef diamond@#(text cap) =

198 inititem@#(cap);

199 z@#lm = z@#bb.lm;

200 z@#mr = z@#bb.mr;

201 z@#um = z@#bb.um;

202 z@#ml = z@#bb.ml;

203 z@#ll = .5[z@#bb.lm,z@#bb.ml];

204 z@#lr = .5[z@#bb.lm,z@#bb.mr];

205 z@#ur = .5[z@#bb.um,z@#bb.mr];

206 z@#ul = .5[z@#bb.um,z@#bb.ml];

207 z@#lc = z@#c;

208 if known z@#ls: z@#ds = begingroup

209 save xt, yt; numeric xt, yt;

210 (xt,yt) = .5z@#ls + if x@#ls>y@#ls: (0,ilmargin)

211 else: (ilmargin,0) fi;

212 2*((xt,yt) + sqrt(xt*yt)*(1,1))

213 endgroup; fi

214 finitem@#(z@#lm--z@#mr--z@#um--z@#ml--cycle)

215 enddef;

oval The default size for ovals is chosen so as to minimize the quantity a2+b2 while still
circumscribing the label, where a and b are half the lengths of the horizontal and
vertical axes, respectively. This avoids highly eccentric ovals in favor of rounder
ones. If both a and b are free, it turns out that the optimal oval satisfies a =√
x(x+ y) and b =

√
y(x+ y), where (x, y) is the upper-right corner of the label

when centered at the origin.

216 vardef oval@#(text cap) =

217 inititem@#(cap);

218 z@#lm = z@#bb.lm;

219 z@#mr = z@#bb.mr;

220 z@#um = z@#bb.um;

221 z@#ml = z@#bb.ml;

222 z@#ll-z@#bb.ll = z@#bb.ur-z@#ur = .5*(1-sqrt(.5))*z@#s;

223 z@#lr-z@#bb.lr = z@#bb.ul-z@#ul = .5*(1-sqrt(.5))*(-x@#s,y@#s);

224 z@#lc = z@#c;

225 if known z@#ls: z@#ds = begingroup

226 save xt,yt; numeric xt,yt;

227 (xt,yt) = .5z@#ls + if x@#ls>y@#ls: (0,ilmargin)

228 else: (ilmargin,0) fi;

229 2*sqrt(xt+yt)*(sqrt(xt),sqrt(yt))

230 endgroup; fi

231 finitem@#(fullcircle xscaled x@#s yscaled y@#s shifted z@#c)

232 enddef;

circ Define a circular item.

233 vardef circ@#(text cap) =

234 inititem@#(cap);

235 (x@#s,0) = (y@#s,0);

236 z@#lm = z@#bb.lm;

237 z@#mr = z@#bb.mr;

238 z@#um = z@#bb.um;

239 z@#ml = z@#bb.ml;

240 z@#ll-z@#bb.ll = z@#bb.ur-z@#ur = .5*(1-sqrt(.5))*z@#s;

241 z@#lr-z@#bb.lr = z@#bb.ul-z@#ul = .5*(1-sqrt(.5))*(-x@#s,y@#s);

242 z@#lc = z@#c;

243 if known z@#ls:

244 z@#ds = length(z@#ls + if x@#ls>y@#ls: (2ilmargin,0)

245 else: (0,2ilmargin) fi) * (1,1);

246 fi

247 finitem@#(fullcircle scaled x@#s shifted z@#c)

248 enddef;

drum Define a drum item. This is currently the only item that does not have a cyclic
path for its frame. (The last full circle draws the lid, and does not end at the
starting point.) To support fill operators (e.g., filledwith), acyclic paths must
start with a cycle (which gets filled) and then finish off with a tail (which is
ignored during filling). We therefore draw the outer border of the drum first and
then finish off with a tail that draws the front of the top lid edge.

249 vardef drum@#(text cap) =

250 inititem@#(cap);

251 z@#lm = z@#bb.lm;

252 z@#mr = z@#bb.mr;

253 z@#um = z@#bb.um;

254 z@#ml = z@#bb.ml;

255 z@#ll-z@#bb.ll = z@#lr-z@#bb.lr = z@#bb.ur-z@#ur =

256 z@#bb.ul-z@#ul = 1.5*(z@#c-z@#lc) = (0,.5drumlidratio*x@#s);

257 z@#ds = z@#ls + (2ilmargin, 2ilmargin + 1.5drumlidratio*x@#ls);

258 finitem@#(

259 z@#ul--(halfcircle xscaled -x@#s yscaled (-drumlidratio*x@#s)

260 shifted .5[z@#ll,z@#lr])--

261 (fullcircle xscaled x@#s yscaled (drumlidratio*x@#s)

262 shifted .5[z@#ul,z@#ur])

263 )

264 enddef;

tornbox A box with a wavy bottom edge.

265 vardef tornbox@#(text cap) =

266 inititem@#(cap);

267 interim truecorners := 1;

268 save p,b; path p; numeric b;

269 p = origin{dir -25}..{right}(1,0);

270 b = ypart (llcorner p);

271 z@#ul = z@#bb.ul;

272 z@#ur = z@#bb.ur;

273 z@#ll = z@#bb.ll - x@#s*(0,b);

274 z@#lr = (x@#bb.lr,y@#ll);

275 z@#ml = .5[z@#ll,z@#ul];

276 z@#mr = .5[z@#lr,z@#ur];

277 z@#um = .5[z@#ul,z@#ur];

278 z@#lm = z@#ll + x@#s*(p intersectionpoint ((.5,0)--(.5,b)));

279 z@#lc = .5[z@#ml,z@#mr];

280 z@#ds = z@#ls + (2ilmargin,2ilmargin-x@#ls*b);

281 finitem@#(z@#ll{dir -25}..{right}z@#lr--z@#ur--z@#ul--cycle)

282 enddef;

drawopen Any shape can be “drawn” without its border by using the drawopen command
instead of draw. The implementation simply evaluates and discards its argument.

283 def drawopen expr p = enddef;

brace Draw a curly brace. This is not a shape like the others since it has different anchor
points and is intrinsically rotatable (without rshape). Therefore, it has its own
specialized implementation.

284 vardef brace@#(expr o) =

285 save t,u,v,w,bh,ew,ret,h,p;

286 numeric t,w,bh,ew,ret,h;

287 pair u,v;

288 path p;

289 z@#d = t[z@#a,z@#b];

290 u = unitvector (z@#b-z@#a);

291 v = u rotated (if ypart(o rotated -angle u)>0: 90 else: -90 fi);

292 z@#c = z@#d + h*v;

293 if unknown t: t=.5; fi

294 if unknown h: h=braceheight; fi

295 if h<0: v:=-v; h:=-h; fi

296 w = min(length(z@#d-z@#a),length(z@#b-z@#d));

297 bh = min(beekheight, h/2);

298 ew = min(clawwidth, w/2);

299 ret = braceret/(2clawwidth)*max(0,min(2clawwidth,w-2clawwidth));

300 p = ( % top-left

301 z@#c{-v rotated -beekangle/2} ..

302 {-u}(z@#c -ew*u -(bh+ret/2-bracethick/2)*v){-u} ..

303 {-u}(z@#a +ew*u +(h-bh+ret/2+bracethick/2)*v){-u} ..

304 {-v}z@#a) &

305 ( % bottom-left

306 z@#a{v rotated -beekangle/2} ..

307 {u}(z@#a +ew*u +(h-bh+ret/2-bracethick/2)*v){u} ..

308 {u}(z@#c -ew*u -(bh+ret/2+bracethick/2)*v){u} ..

309 {v}(z@#c -.5bracethick*v)) &

310 ( % bottom-right

311 (z@#c -.5bracethick*v){-v} ..

312 {u}(z@#c +ew*u -(bh+ret/2+bracethick/2)*v){u} ..

313 {u}(z@#b -ew*u +(h-bh+ret/2-bracethick/2)*v){u} ..

314 {-v rotated beekangle/2}z@#b) &

315 ( % top-right

316 z@#b{v} ..

317 {-u}(z@#b -ew*u +(h-bh+ret/2+bracethick/2)*v){-u} ..

318 {-u}(z@#c +ew*u -(bh+ret/2-bracethick/2)*v){-u} ..

319 {v rotated beekangle/2}z@#c) & cycle;

320 if (path ip@#) and (unknown ip@#): ip@#=p; ip@# else: p fi

321 enddef;

inback Execute a drawing command so as to put its results behind everything that has
already been drawn.

322 def inback text t =

323 begingroup

324 save pic_, ils_;

325 picture pic_,ils_;

326 pic_ = currentpicture;

327 ils_ = itemlabels;

328 currentpicture := nullpicture;

329 itemlabels := nullpicture;

330 t;

331 addto currentpicture also itemlabels;

332 addto currentpicture also pic_;

333 itemlabels := ils_;

334 endgroup

335 enddef;

turntolayer Switch to a different (possibly already existing) layer.

336 vardef turntolayer(expr n) =

337 save t; numeric t;

338 addto currentpicture also itemlabels;

339 itemlabels := nullpicture;

340 layer[thislayer] := currentpicture;

341 thislayer := n;

342 currentpicture := if known layer[n]: layer[n] else: nullpicture fi;

343 layer[n] := nullpicture;

344 t = xpart (layerlist intersectiontimes (n,n));

345 if t=-1:

346 layerlist := if ((n,n)<point 0 of layerlist): (n,n)..layerlist

347 else: layerlist..(n,n) fi;

348 elseif t>floor t:

349 t := floor t;

350 layerlist := (subpath (0,t) of layerlist)..(n,n)..

351 (subpath (t+1,length layerlist) of layerlist);

352 fi

353 enddef;

flatten Flatten all layers onto layer 0, and make it current.

354 def flatten =

355 addto currentpicture also itemlabels;

356 if length layerlist>0:

357 layer[thislayer] := currentpicture;

358 currentpicture := nullpicture;

359 for t=0 upto length layerlist:

360 addto currentpicture also layer[xpart point t of layerlist];

361 endfor

362 picture layer[];

363 thislayer := 0;

364 layerlist := origin;

365 fi

366 enddef;

anchor Convert a direction vector to an anchor point name.

367 def anchor(suffix $)(expr d) =

368 (if (xpart d)=0:

369 if (ypart d)=0: $c elseif (ypart d)>0: $um else: $lm fi

370 elseif (xpart d)>0:

371 if (ypart d)=0: $mr elseif (ypart d)>0: $ur else: $lr fi

372 elseif (ypart d)=0: $ml elseif (ypart d)>0: $ul else: $ll fi)

373 enddef;

upright

downright

upleft

downleft

It is helpful to have unit vectors for the ordinal directions in addition to the
cardinal ones provided by plain METAPOST.

374 pair upright, downright, upleft, downleft;

375 upright = -downleft = unitvector (up+right);

376 downright = -upleft = unitvector (down+right);

putitem Position an item relative to another. If 〈d〉 is a vector in a cardinal direction, a
constraint is introduced that separates the relevant opposing bounding box mid-
points by 〈d〉. Otherwise the constraint is between opposing bounding box corner
points.

377 vardef putitem[] expr d of i =

378 anchor(z@bb,-d) = (if pair i: i else: anchor(z[i]bb,d) fi) + d

379 enddef;

putitems

like

Position one pair of items like another pair of items. To get the custom syntax
putitems(i,j) like(i′,j′), we define macro putitems so that it herds its two
arguments into a 4-argument like macro. In order to introduce the proper con-
straint, at least one of the two item pairs must have fully known relative positions.
The known pair can be safely examined without raising an unresolved constraint
error. Based on the results, we introduce new (possibly heretofore unresolved)
constraints for the other pair.

380 def putitems(suffix $,$$) text t = t($,$$) enddef;

381 vardef like(suffix $,$$,#,##) =

382 save d; pair d;

383 d = if known (z$$c-z$c): z$$c-z$c else: z##c-z#c fi;

384 anchor(z##,-d) - anchor(z#,d) = anchor(z$$,-d) - anchor(z$,d);

385 enddef;

_corners

urcorners

ulcorners

llcorners

lrcorners

Compute the corners of a set of objects.

386 vardef _corners(text #,##,op)(expr u)(text t) =

387 interim truecorners := 1;

388 save ux_,uy_,v_; numeric ux_,uy_; pair v_;

389 (ux_,uy_) = if pair u: u else: op u fi;

390 for uu = t:

391 v_ := if pair uu: uu else: op uu fi;

392 if (xpart v_)#ux_: ux_:=xpart v_; fi

393 if (ypart v_)##uy_: uy_:=ypart v_; fi

394 endfor

395 (ux_,uy_)

396 enddef;

397 def urcorners = _corners(>)(>)(urcorner) enddef;

398 def ulcorners = _corners(<)(>)(ulcorner) enddef;

399 def llcorners = _corners(<)(<)(llcorner) enddef;

400 def lrcorners = _corners(>)(<)(lrcorner) enddef;

filledwith Operation (〈p〉 filledwith 〈f 〉) fills a path 〈p〉 with a color or picture 〈f 〉. If
path 〈p〉 is acyclic, we look for a cyclic prefix subpath. (This works for drum
shapes—the only acyclic shape at present.) If there is none, we just close it to
make it a cycle and hope that works.

If 〈f 〉 is a color, a solid fill is contributed. If it is a picture, it is simply clipped
to the path without any centering or tesselation. The other fill operators use this
latter functionality to contribute their fill patterns.

401 tertiarydef p filledwith f =

402 begingroup

403 save c; path c;

404 c = (if picture p:

405 begingroup interim truecorners := 1;

406 bbox p

407 endgroup

408 elseif cycle p: p

409 else:

410 for t=1 upto length p:

411 if point t of p = point 0 of p:

412 (subpath (0,t) of p) & cycle

413 elseif t = length p: p..cycle fi

414 exitif point t of p = point 0 of p;

415 endfor

416 fi);

417 if color f:

418 fill c withcolor f;

419 elseif picture f:

420 save pic;

421 picture pic;

422 pic = f;

423 clip pic to c;

424 draw pic;

425 else:

426 errmessage("non-color/picture argument to filledwith ignored");

427 fi;

428 p

429 endgroup

430 enddef;

tesselatedwith Create a fill pattern by tesselating a rectangular picture. The tesselation is aligned
with the original picture’s location, not any reference point of the shape it fills, so
that nearby shapes with the same tesselated fill pattern look like windows into an
unbroken underlying pattern. This makes nearby shapes with the same fill pattern
look more compatible.

431 tertiarydef b tesselatedwith p =

432 begingroup

433 save tpic, pic, llx, lly, urx, ury, psizx, psizy;

434 picture tpic, pic;

435 tpic := nullpicture;

436 pic = p;

437 (psizx,psizy) = (urcorner pic) - (llcorner pic);

438 (llx,lly) = (llcorner pic) + ((llcorner b) - (llcorner pic));

439 llx := llx div psizx * psizx;

440 lly := lly div psizy * psizy;

441 (urx,ury) = (urcorner b) + (psizx,psizy);

442 for i = llx step psizx until urx:

443 for j = lly step psizy until ury:

444 addto tpic also (pic shifted (i,j));

445 endfor;

446 endfor;

447 b filledwith tpic

448 endgroup

449 enddef;

stripedwith Operation (〈b〉 stripedwith 〈p〉) fills a bounding path 〈b〉 with a stripe tesselation
obtained by projecting every line segment in picture 〈p〉 orthogonally to form a
stripe. Zero-length line segments (i.e., points) are directionless, so in that case the
projection is orthogonal to the direction of the first line segment in the picture,
the second line segment if the first one is zero-length, or the direction from the
first to the second if the first two are both zero-length. A picture consisting of a
single, zero-length line segment is ignored, yielding an empty pattern. All non-
lines in the picture are also ignored. Colors of line segments are preserved, allowing
multicolored patterns. Pen styles of zero-length line segments are also preserved
(since those are projected to lines, for which a pen style makes sense).

450 tertiarydef b stripedwith p =

451 begingroup

452 save tpic, pic, pl, e, d, s, r, dl, fp, ll, ur, x, gr;

453 picture tpic, pic;

454 numeric pl, dl, r, gr;

455 pair s, ll, ur;

456 path d, fp;

457 tpic := nullpicture;

458 pic = p;

459 pl = length ((urcorner pic)-(llcorner pic));

460 for e within pic:

461 if stroked e:

462 d := pathpart e;

463 if ((point 0 of d) <> (point infinity of d)):

464 gr = angle ((point infinity of d)-(point 0 of d));

465 elseif (point 0 of (pathpart pic)) <> (point 0 of d):

466 gr = angle ((point 0 of d)-(point 0 of (pathpart pic)));

467 fi;

468 fi;

469 exitif known gr;

470 endfor;

471 if (known gr) and (pl>0):

472 for e within pic:

473 if stroked e:

474 d := pathpart e;

475 s := point 0 of d;

476 dl := length ((point infinity of d)-s);

477 r := if dl>0: angle ((point infinity of d)-s) else: gr fi;

478 fp := b shifted -s rotated -r;

479 ll := llcorner fp;

480 ur := urcorner fp;

481 for x = ((xpart ll) div pl*pl) step pl until (xpart ur)+pl:

482 if dl>0:

483 addto tpic contour ((x,ypart ll)--(x+dl,ypart ll)--

484 (x+dl,ypart ur)--(x,ypart ur)--cycle)

485 rotated r shifted s withcolor (colorpart e);

486 else:

487 addto tpic doublepath ((x,ypart ll)--(x,ypart ur))

488 rotated r shifted s

489 withpen (penpart e) withcolor (colorpart e);

490 fi;

491 endfor;

492 fi;

493 endfor;

494 fi;

495 b filledwith tpic

496 endgroup

497 enddef;

dashstripes Convert a dash pattern to a stripe pattern. Dash patterns are almost valid stripe
patterns already, except that they lack proper bounding boxes. METAPOST adopts
the peculiar convention of representing a dash pattern as a horizontal series of line
segments whose position above the y-axis is the total width of the pattern. When
the last part of the pattern is an “off”, the y-position of the pattern will therefore
be larger than the x-position of the last dash’s endpoint. To construct a correct
bounding box, we therefore just compute the max of the x- and y-positions. The

addition of the bounding box allows the stripe pattern to be properly tesselated
after rotation, since rotated stripe patterns are no longer horizontal lines with
fixed y-positions.

498 vardef dashstripes primary p =

499 save pic, ur;

500 picture pic; pic = p;

501 pair ur; ur = urcorner pic;

502 setbounds pic to

503 (ulcorner pic)--(max(xpart ur,ypart ur), ypart ur)--cycle;

504 pic

505 enddef;

evenstripes

pinstripes

The evenstripes and pinstripes patterns are the stripe analogs of the evenly

and withdots dash patterns. To make them easier for the user to shift and rotate,
we reposition and reorient them so that the stripes run horizontally and are aligned
with the x-axis. The withdots dash pattern cannot be directly used to create
pinstripes because it has only one, zero-length dash, making it directionless.
We must therefore construct a doubled version of the withdots pattern so that it
has two dashes.

506 picture evenstripes, pinstripes;

507 evenstripes = dashstripes evenly shifted -(ulcorner evenly)

508 rotated 90;

509 pinstripes = dashstripes dashpattern(on 0 off 5 on 0 off 5)

510 shifted -(0,10) rotated 90;

colored Operation (〈p〉 colored 〈c〉) recolors a picture 〈p〉 a new color 〈c〉. This is useful
for changing the color of stripe patterns.

511 primarydef p colored c =

512 begingroup

513 save pic;

514 picture pic;

515 pic := nullpicture;

516 addto pic also p withcolor c;

517 pic

518 endgroup

519 enddef;

onhue To make multicolor stripe patterns, we need a dash pattern constructor like on

except with an extra color parameter. METAPOST does not have any means of
defining parameterized binary operators, so to immitate one, we first define macro
onhue(〈c〉)〈d〉 so that it creates a picture of a 〈c〉-colored, length-〈d〉 dash, and
then expands to binary operator _onhue_ applied to that picture.

520 def onhue(expr c) secondary d =

521 _onhue_

522 begingroup save pic;

523 picture pic; pic=nullpicture;

524 addto pic doublepath (0,d)..(d,d) withcolor c;

525 pic

526 endgroup

527 enddef;

_onhue_ Binary operation (〈p〉 _onhue_ 〈d〉) adds dash 〈d〉 to picture 〈p〉. This is essentially
like on except that 〈d〉 is an entire picture, not just a numeric length.

528 tertiarydef p _onhue_ d =

529 begingroup save pic, ur, delta;

530 picture pic; pic=p;

531 pair ur; ur=urcorner d;

532 numeric delta; delta=max(xpart ur,ypart ur);

533 addto pic also d shifted ((w,w)-(llcorner d));

534 w := w+delta;

535 pic shifted (0,delta)

536 endgroup

537 enddef;

connector Return a connector path exiting item 〈$〉 in direction 〈dsrc〉 and entering item
〈$$〉 in direction 〈ddst〉. If the connector is unnamed, give it a temporary name.

538 vardef connector@#(suffix $,$$)(expr dsrc,ddst) =

539 if (str @#)="":

540 numeric x[]cp.tmp, y[]cp.tmp;

541 path cp.tmp;

542 _connector.tmp

543 else:

544 if known cp@#:

545 errmessage("redundant connector name: " & (str @#));

546 fi;

547 _connector@#

548 fi($,$$,dsrc,ddst)

549 enddef;

There are 16 cases that must be considered for connector paths—one for each
exit-entry cardinal direction pair. We can reduce this to 4 cases by first rotating
everything so that the exit direction is rightward, solving the resulting connector
path problem, and then re-rotating back to the original orientation. This strategy
reduces the set of possibilities to the 4 possible entry directions.

Rather than doing the rotation using polar coordinates, which would entail
non-linear constraints that METAPOST cannot solve automatically, we formulate
the rotation as a reflection and/or juxtaposition of x- and y-ordinates. For exam-
ple, mapping upward to rightward can be achieved by a juxtaposition and then
an x-reflection.

_jux This helper macro conditionally juxtaposes and possibly inverts axes in a con-
straint in order to rotate everything so that the exit direction of the connector is
rightward.

550 def _jux(text a,b) =

551 if s.h=s.v: ((a)*s.h,(b)*s.v) else: ((b)*s.v,(a)*s.h) fi

552 enddef;

_iv This macro chooses amongst 4 choices based on angle a. The choices are for up,
left, down, and right, respectively.

553 def _iv(expr a)(suffix b,c,d,e) =

554 if (45 <= a) and (a < 135): b

555 elseif (135 <= a) and (a < 225): c

556 elseif (225 <= a) and (a < 295): d

557 else: e fi

558 enddef;

_connector Rotate a connector path problem so that the exit direction is rightward, and then
invoke the appropriate sub-logic for the appropriate (rotated) entry direction.
Variables i〈n〉r, i〈n〉l, i〈n〉t, and i〈n〉b store the right, left, top, and bottom
ordinates (respectively) of the source (n = 0) and destination (n = 1) items after
rotation. Variables s.h and s.v store −1 if the horizontal or vertical direction
(respectively) is being reflected after rotation, and 1 otherwise.

559 vardef _connector@#(suffix $,$$)(expr dsrc,ddst) =

560 save i, s;

561 numeric i[]a, i[]r, i[]l, i[]t, i[]b, i[]x, i[]y, s.h, s.v;

562 i0a = (angle dsrc) mod 360;

563 i1a = (angle -ddst) mod 360;

564 s.h = (if (135 <= i0a) and (i0a < 295): -1 else: 1 fi);

565 s.v = (if (45 <= i0a) and (i0a < 225): -1 else: 1 fi);

566 _jux(i0l)(i0b) = z$bb _iv(i0a,lr,ur,ul,ll);

567 _jux(i0r)(i0t) = z$bb _iv(i0a,ul,ll,lr,ur);

568 _jux(i1l)(i1b) = z$$bb _iv(i0a,lr,ur,ul,ll);

569 _jux(i1r)(i1t) = z$$bb _iv(i0a,ul,ll,lr,ur);

570 _jux(i0x)(i0y) = z$ _iv(i0a,um,ml,lm,mr);

571 _jux(i1x)(i1y) = z$$ _iv(i1a,um,ml,lm,mr);

572 _iv((i1a-i0a+360) mod 360,

573 _conn_down,_conn_right,_conn_up,_conn_left)@#

574 enddef;

_conpath Each _conn_〈dir〉 macro (below) concludes with a call to the following macro,
which re-rotates back to the original exit direction and returns the resulting con-
nector path. The input to the macro is the suggested series of alternating x-
and y-ordinates for the (rotated) path. Each ordinate is overridden with a user-
supplied choice if it has already been defined by the user.

575 vardef _conpath@#(text tail) =

576 save n,h;

577 numeric n;

578 boolean h;

579 h := (s.h = s.v);

580 if unknown x0cp@#: x0cp@# = (if h: i0x*s.h else: i0y*s.v fi) fi;

581 if unknown y0cp@#: y0cp@# = (if h: i0y*s.v else: i0x*s.h fi) fi;

582 n := 0;

583 for o=tail:

584 if h:

585 if unknown y[n+1]cp@#: y[n+1]cp@# = y[n]cp@#; fi;

586 if unknown x[n+1]cp@#:

587 x[n+1]cp@#*(if odd n: s.v else: s.h fi) = o; fi;

588 else:

589 if unknown x[n+1]cp@#: x[n+1]cp@# = x[n]cp@#; fi;

590 if unknown y[n+1]cp@#:

591 y[n+1]cp@#*(if odd n: s.v else: s.h fi) = o; fi;

592 fi;

593 h := not h;

594 n := n+1;

595 endfor;

596 if (unknown cp@#) and (numeric cp@#): path cp@#; fi;

597 cp@# = z0cp@# for j=1 upto n: --z[j]cp@# endfor;

598 cp@#

599 enddef;

The following macros solve the connector path problem for each of the possible
entry directions. They all assume that the exit direction is rightward.

_conn_right Compute a right-exiting, right-entering connector path.

600 vardef _conn_right@# =

601 if (i0y=i1y) and (i0x <= i1x):

602 _conpath@#(i1x)

603 elseif (i0r+2cmargin <= i1l) or

604 ((i0x <= i1x) and

605 (i1b < i0t+2cmargin) and (i1t > i0b-2cmargin)):

606 _conpath@#(.5[i0r,i1l],i1y,i1x)

607 elseif (i1b >= i0t+2cmargin) or (i1t <= i0b-2cmargin):

608 _conpath@#(i0r+cmargin,.5[i0t,i1b],i1l-cmargin,i1y,i1x)

609 elseif (i1y <= i0y):

610 _conpath@#(i0r+cmargin,min(i0b,i1b)-cmargin,i1l-cmargin,i1y,i1x)

611 else:

612 _conpath@#(i0r+cmargin,max(i0t,i1t)+cmargin,i1l-cmargin,i1y,i1x)

613 fi

614 enddef;

_conn_up Compute a right-exiting, up-entering connector path.

615 vardef _conn_up@# =

616 if (i1l >= i0r+2cmargin) and (i1y < i0y+cmargin):

617 _conpath@#(.5[i0r,i1l],i1b-cmargin,i1x,i1y)

618 elseif (i1y < i0y) or

619 ((i1x < i0l) and (i1y <= i0t+2cmargin)):

620 _conpath@#(max(i0r,i1r)+cmargin,min(i0b,i1b)-cmargin,i1x,i1y)

621 elseif (i1x <= i0r) or

622 ((i1x < i0r+cmargin) and (i1b >= i0t+2cmargin)):

623 _conpath@#(i0r+cmargin,.5[i0t,i1b],i1x,i1y)

624 else:

625 _conpath@#(i1x,i1y)

626 fi

627 enddef;

_conn_down Compute a right-exiting, down-entering connector path.

628 vardef _conn_down@# =

629 if (i1l >= i0r+2cmargin) and (i1y > i0y-cmargin):

630 _conpath@#(.5[i0r,i1l],i1t+cmargin,i1x,i1y)

631 elseif (i1y > i0y) or

632 ((i1x < i0l) and (i1y <= i0b-2cmargin)):

633 _conpath@#(max(i0r,i1r)+cmargin,max(i0t,i1t)+cmargin,i1x,i1y)

634 elseif (i1x <= i0r) or

635 ((i1x < i0r+cmargin) and (i1b <= i0b-2cmargin)):

636 _conpath@#(i0r+cmargin,.5[i0b,i1t],i1x,i1y)

637 else:

638 _conpath@#(i1x,i1y)

639 fi

640 enddef;

_conn_left Compute a right-exiting, left-entering connector path.

641 vardef _conn_left@# =

642 if (i1x <= i0l-2cmargin) and

643 (i1y <= .5[i0b,i0t]) and (i1y > i0b-cmargin):

644 _conpath@#(i0r+cmargin,i0b-cmargin,.5[i0l,i1r],i1y,i1x)

645 elseif (i1x <= i0l-2cmargin) and

646 (i1y > .5[i0b,i0t]) and (i1y < i0t+cmargin):

647 _conpath@#(i0r+cmargin,i0t+cmargin,.5[i0l,i1r],i1y,i1x)

648 elseif (i1l >= i0r+2cmargin) and

649 (i1b < i0y+cmargin) and (i1t > i0y-cmargin):

650 if (abs(i1t-i0y) < abs(i0y-i1b)):

651 _conpath@#(.5[i0r,i1l],i1t+cmargin,i1r+cmargin,i1y,i1x)

652 else:

653 _conpath@#(.5[i0r,i1l],i1b-cmargin,i1r+cmargin,i1y,i1x)

654 fi

655 else:

656 _conpath@#(max(i0r,i1r)+cmargin,i1y,i1x)

657 fi

658 enddef;

rshape Create a rotated shape. Known bug: rshape might not work with a non-integer
name.

659 vardef rshape@#(suffix $)(expr d)(text cap) =

660 save a,s;

661 a = (angle d) mod 360;

662 s.h = (if (135 <= a) and (a < 315): -1 else: 1 fi);

663 s.v = (if (45 <= a) and (a < 225): 1 else: -1 fi);

664 _jux(x@#c)(y@#c) = z.xf@#c;

665 _jux(x@#lc)(y@#lc) = z.xf@#lc;

666 forsuffixes u=s,ls,ds:

667 z@#u = (if s.h=s.v: z.xf@#u else: (y.xf@#u,x.xf@#u) fi);

668 endfor

669 forsuffixes u=,bb:

670 _jux(x@#u.ul)(y@#u.ul) = z.xf@#u _iv(a,ul,ur,lr,ll);

671 _jux(x@#u.ml)(y@#u.ml) = z.xf@#u _iv(a,ml,um,mr,lm);

672 _jux(x@#u.ll)(y@#u.ll) = z.xf@#u _iv(a,ll,ul,ur,lr);

673 _jux(x@#u.lm)(y@#u.lm) = z.xf@#u _iv(a,lm,ml,um,mr);

674 _jux(x@#u.lr)(y@#u.lr) = z.xf@#u _iv(a,lr,ll,ul,ur);

675 _jux(x@#u.mr)(y@#u.mr) = z.xf@#u _iv(a,mr,lm,ml,um);

676 _jux(x@#u.ur)(y@#u.ur) = z.xf@#u _iv(a,ur,lr,ll,ul);

677 _jux(x@#u.um)(y@#u.um) = z.xf@#u _iv(a,um,mr,lm,ml);

678 endfor

679 inititem@#(cap);

680 if itemfinal:

681 save pth; path pth;

682 pth = $.xf@#() rotated ((a-90) div 90 * 90);

683 finitem@#(pth)

684 else:

685 $.xf@#(false)

686 fi

687 enddef;

supertime Translate a time along a subpath to a time along its superpath. That is, if
p′ = subpath (t1, t2) of p and t = supertime t′ of (t1, t2), then point t of p =
point t′ of p′.

688 vardef supertime expr t of b =

689 save s,e; (s,e) = b;

690 if t<=-1: t

691 elseif t<=1:

692 t[s,if e<s: max(ceiling(s)-1,e) else: min(floor(s)+1,e) fi]

693 elseif t<=abs(floor(e)-floor(s)):

694 if e<s: ceiling(s)-t else: floor(s)+t fi

695 elseif t<abs(floor(e)-floor(s))+1:

696 (t-floor(t))[if e<s: ceiling(e) else: floor(e) fi,e]

697 else: t fi

698 enddef;

__poppath Reserve a global array for storing arrays of paths used in computing popovers.

699 path __poppath[];

popover The top-level popover macro has syntax like a binary operator, but its second
argument is a list of paths, which is not a legal data type in METAPOST. We
therefore evaluate and store the paths into a path array first, and then expand
to a real binary operator. Note that each path expression might itself contain a
popover macro, so some careful grouping is required.

700 def popover(text pths) =

701 _popover begingroup

702 save __n;

703 __n:=0;

704 for x=pths:

705 __poppath[__n] = begingroup save __poppath; x endgroup;

706 __n := __n + 1;

707 endfor

708 __n

709 endgroup

710 enddef;

_popover Next, a special case is required for cycles. If a cycle has an intersection near its
endpoints, it is first re-parameterized to shift its endpoint away from the intersec-
tion. This allows the rest of the code to safely treat the path as a non-cycle.

711 tertiarydef p _popover n =

712 if cycle p: begingroup

713 save t,u,c,q,r,s; path c,q,r,s;

714 c = fullcircle scaled 2pradius shifted point 0 of p;

715 t = xpart (p intersectiontimes c);

716 u = xpart ((reverse p) intersectiontimes c);

717 if (t<0) or (u<0): p else:

718 q = subpath (0,t) of p;

719 r = subpath (0,u) of reverse p;

720 for i=0 upto n-1:

721 if xpart(q intersectiontimes __poppath[i])>=0:

722 s = __popover(subpath (-u,length p - u) of p,n) -- cycle;

723 elseif xpart(r intersectiontimes __poppath[i])>=0:

724 s = __popover(subpath (t,length p + t) of p,n) -- cycle;

725 fi

726 exitif known s;

727 endfor

728 if known s: s else: __popover(p,n) & cycle fi

729 fi

730 endgroup else: __popover(p,n) fi

731 enddef;

__popover The following macro returns a new path like p except spliced with circular arcs
of radius pradius everywhere p intersects one of the n paths in the __poppath

array. Intersections closer than pradius to one another or to the ends of the path
are ignored. The pops try to prefer upward and rightward pop directions except
when p is bent at the point of intersection. In that case, the pops take the “long
way” around the circle to maximize visibility.

732 vardef __popover(expr p,n) =

733 save t;

734 t := -1;

735 for i=0 upto n-1:

736 t := xpart(p intersectiontimes __poppath[i]);

737 exitif t>=0;

738 endfor

739 if t<=0: p else:

740 save i,c,st,et,sv,ev,a; pair i,sv,ev; path c;

741 i = point t of p;

742 c = fullcircle scaled 2pradius shifted i;

743 st = xpart ((subpath (t,0) of p) intersectiontimes c);

744 et = xpart ((subpath (t,length p) of p) intersectiontimes c);

745 if (st<0) or (et<0): p else:

746 st := supertime st of (t,0);

747 et := supertime et of (t,length p);

748 sv = point st of p - i;

749 ev = point et of p - i;

750 a = angle(ev rotated -angle sv);

751 __popover(subpath (0,st) of p, n) --

752 (if (abs(a)>=179):

753 if (-91<angle ev) and (angle ev<89): reverse fi

754 elseif a>0: reverse fi

755 fullcircle zscaled 2sv cutafter (origin--2ev)) shifted i --

756 __popover(subpath (et,length p) of p, n)

757 fi

758 fi

759 enddef;

cfilldraw Create a macro like filldraw except that it draws without filling when its argu-
ment is an acyclic path.

760 def cfilldraw expr p =

761 addto currentpicture

762 if cycle p: contour else: doublepath fi p

763 withpen currentpen _op_

764 enddef;

_finarr This replaces the filldraw commands in the drawarrow macro with cfilldraw,
so that the arrowhead macro may return an acyclic path that is simply drawn,
not filled.

765 vardef _finarr text t =

766 draw _apth t;

767 cfilldraw arrowhead _apth t

768 enddef;

_findarr Likewise, this replaces the filldraw commands in the drawdblarrow macro with
cfilldraw.

769 vardef _findarr text t =

770 draw _apth t;

771 cfilldraw arrowhead _apth withpen currentpen t;

772 cfilldraw arrowhead reverse _apth withpen currentpen t;

773 enddef;

The default METAPOST code for arrowheads has an aesthetic flaw that we here
correct. It computes the front edge of an arrowhead for path p by rotating the
subpath q of points within distance ahlength of p’s endpoint both ahangle/2 de-
grees clockwise and counter-clockwise, forming a pointed vee. This works when the
path is linear, but when it’s curved, two problems arise: (1) Subpath q is slightly
too long—the distance along q is greater than ahlength (though the straight-line
chord from its start point to endpoint is indeed ahlength). (2) The back edge of
the arrowhead is not orthogonal to p where they intersect, making the arrowhead
look noticeably lopsided (see Fig. 14(a)).

(a) A default METAPOST arrowhead (b) A METAf low arrowhead

Figure 14: Arrowheads before and after correction

A correct arrowhead (see Fig. 14(b)) should instead be the result of projecting
each point t orthogonally to the direction of p at t. Thus, the half of the arrowhead
projected outside curve p should be longer than the half projected inside the
curve, making the straight connecting line exactly orthogonal to p. In general, the
direction of the arrowhead’s edge at each point t should be the direction of p at
point t rotated ahangle/2 degrees inward toward p.

The exact formula for this curve is not generally representable as a cubic Bézier
curve (see research on offset curves), but it can be reasonably approximated by
computing the proper points and trajectories at integer times t and letting META-
POST interpolate the rest.

taper Compute a new path that tapers toward path 〈p〉 until it intersects 〈p〉 at its
endpoint forming angle 〈a〉.

774 vardef taper(expr p,a) =

775 save r;

776 numeric r;

777 r = sind(a)/cosd(a);

778 (point 0 of p +

779 r * arclength p * unitvector direction 0 of p rotated 90)

780 {(direction 0 of p) rotated -a}

781 for t=1 upto (length p)-1:

782 .. {(point t of p - precontrol t of p) rotated -a}

783 (point t of p +

784 r * (arclength subpath (t,length p) of p) *

785 dir (.5[angle (point t of p - precontrol t of p),

786 angle (postcontrol t of p - point t of p)] + 90))

787 {(postcontrol t of p - point t of p) rotated -a}

788 endfor

789 .. {(direction length p of p) rotated -a}(point length p of p)

790 enddef;

sarrowhead A straight arrowhead has a straight line for its back edge.

791 vardef sarrowhead expr p =

792 save q; path q;

793 q = subpath (arctime (arclength p - ahlength) of p,length p) of p;

794 (taper(q,.5ahangle) &

795 reverse taper(q,-.5ahangle) -- cycle)

796 enddef;

oarrowhead An open arrowhead is unfilled.

797 vardef oarrowhead expr p =

798 save q; path q;

799 q = subpath (arctime (arclength p - ahlength) of p,length p) of p;

800 (taper(q,.5ahangle) &

801 reverse taper(q,-.5ahangle))

802 enddef;

varrowhead A V-arrowhead insets the back with a V-shape.

803 vardef varrowhead expr p =

804 save va,q,qq;

805 numeric va;

806 path q,qq;

807 va = angle (ahinset*ahlength,

808 ahlength*sind(.5ahangle)/cosd(.5ahangle));

809 q = subpath (arctime (arclength p - ahlength) of p,length p) of p;

810 qq = subpath (0,arctime ahinset*ahlength of q) of q;

811 (reverse taper(qq,va) ..

812 taper(q,.5ahangle) &

813 reverse taper(q,-.5ahangle) ..

814 taper(qq,-va) & cycle)

815 enddef;

ahinset The depth of the V in a V-arrow is determined by the value of ahinset which
should be a number between 0 and 1. The larger the number, the deeper the V.

816 newinternal ahinset;

817 ahinset := 0;

arrowhead Replace METAPOST’s default arrowhead macro with one that chooses the arrow-
head type based on the value of ahinset.

818 vardef arrowhead expr p =

819 if ahinset <= 0: sarrowhead

820 elseif ahinset >= 1: oarrowhead

821 else: varrowhead fi p

822 enddef;

